<commit_before>SUMMARY = "Soundfont editor"
HOMEPAGE = "http://polyphone-soundfonts.com/en/"
LICENSE = "GPLv3"
LIC_FILES_CHKSUM = "file://main.cpp;beginline=6;endline=18;md5=092eabd5b12b1c1d876a94e06601bd81"

inherit qmake5 gtk-icon-cache mime

DEPENDS += " \
    qtbase \
    qtsvg \
    alsa-lib \
    jack \
    portaudio-v19 \
    rtmidi \
    qcustomplot \
    stk \
    libvorbis \
    libogg \
    unzip-native \
"

SRC_URI = " \
    http://polyphone-soundfonts.com/en/download/file/76-polyphone-1-8-src-zip/latest/download;downloadfilename=polyphone-1-8-src.zip \
    file://polyphone.desktop \
    file://polyphone.mime \
    file://0001-align-compiler-switches-constants-for-cross-compilin.patch \
"
SRC_URI[md5sum] = "75ec14ab98bcf80f360debd80c553733"
SRC_URI[sha256sum] = "2b0ba2887646208ae9953a336adfebe460133cd403e76f7fe795d3071827dfdd"

S = "${WORKDIR}/trunk"

do_configure_prepend() {
    sed -i 's:= /usr/include:= ${STAGING_INCDIR}:g' ${S}/polyphone.pro
}

do_install() {
    install -d ${D}${bindir}
    install -m 0755 ${B}/RELEASE/polyphone ${D}${bindir}/

    install -d ${D}${datadir}/pixmaps
    install -m 0644 ${S}/ressources/polyphone.png ${D}${datadir}/pixmaps/

    install -d ${D}${datadir}/applications
    install -m 0644 ${WORKDIR}/${BPN}.desktop ${D}${datadir}/applications/

    install -d ${D}${datadir}/mime/packages
	install -m 0644 ${WORKDIR}/${BPN}.mime ${D}${datadir}/mime/packages/${BPN}.xml
}

FILES_${PN} += " \
    ${datadir}/mime \
"

<commit_msg>polyphone: work around our do_unzip problem effectively<commit_after>SUMMARY = "Soundfont editor"
HOMEPAGE = "http://polyphone-soundfonts.com/en/"
LICENSE = "GPLv3"
LIC_FILES_CHKSUM = "file://main.cpp;beginline=6;endline=18;md5=092eabd5b12b1c1d876a94e06601bd81"

inherit qmake5 gtk-icon-cache mime

DEPENDS += " \
    qtbase \
    qtsvg \
    alsa-lib \
    jack \
    portaudio-v19 \
    rtmidi \
    qcustomplot \
    stk \
    libvorbis \
    libogg \
"

SRC_URI = " \
    http://polyphone-soundfonts.com/en/download/file/76-polyphone-1-8-src-zip/latest/download;downloadfilename=polyphone-1-8-src.zip \
    file://polyphone.desktop \
    file://polyphone.mime \
    file://0001-align-compiler-switches-constants-for-cross-compilin.patch \
"
SRC_URI[md5sum] = "75ec14ab98bcf80f360debd80c553733"
SRC_URI[sha256sum] = "2b0ba2887646208ae9953a336adfebe460133cd403e76f7fe795d3071827dfdd"

S = "${WORKDIR}/trunk"

# Have no idea why this is necessary
do_unpack[depends] += "unzip-native:do_populate_sysroot"

do_configure_prepend() {
    sed -i 's:= /usr/include:= ${STAGING_INCDIR}:g' ${S}/polyphone.pro
}

do_install() {
    install -d ${D}${bindir}
    install -m 0755 ${B}/RELEASE/polyphone ${D}${bindir}/

    install -d ${D}${datadir}/pixmaps
    install -m 0644 ${S}/ressources/polyphone.png ${D}${datadir}/pixmaps/

    install -d ${D}${datadir}/applications
    install -m 0644 ${WORKDIR}/${BPN}.desktop ${D}${datadir}/applications/

    install -d ${D}${datadir}/mime/packages
	install -m 0644 ${WORKDIR}/${BPN}.mime ${D}${datadir}/mime/packages/${BPN}.xml
}

FILES_${PN} += " \
    ${datadir}/mime \
"


{'fraction_non_alphanumeric': 0.16817010309278352, 'fraction_numerical': 0.07603092783505154, 'mean_word_length': 3.6205357142857144, 'pattern_counts': {'":': 0, '<': 3, '<?xml version=': 0, '>': 3, 'https://': 0, 'lorem ipsum': 0, 'www.': 0, 'xml': 2}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 6, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '371385', 'n_tokens_mistral': 1421, 'n_tokens_neox': 1263, 'n_words': 170}